// File: Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (utils/Address.sol)

pragma solidity ^0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}
// File: Counters.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (utils/Counters.sol)

pragma solidity ^0.8.0;

/**
 * @title Counters
 * @author Matt Condon (@shrugs)
 * @dev Provides counters that can only be incremented, decremented or reset. This can be used e.g. to track the number
 * of elements in a mapping, issuing ERC721 ids, or counting request ids.
 *
 * Include with `using Counters for Counters.Counter;`
 */
library Counters {
    struct Counter {
        // This variable should never be directly accessed by users of the library: interactions must be restricted to
        // the library's function. As of Solidity v0.5.2, this cannot be enforced, though there is a proposal to add
        // this feature: see https://github.com/ethereum/solidity/issues/4637
        uint256 _value; // default: 0
    }

    function current(Counter storage counter) internal view returns (uint256) {
        return counter._value;
    }

    function increment(Counter storage counter) internal {
        unchecked {
            counter._value += 1;
        }
    }

    function decrement(Counter storage counter) internal {
        uint256 value = counter._value;
        require(value > 0, "Counter: decrement overflow");
        unchecked {
            counter._value = value - 1;
        }
    }

    function reset(Counter storage counter) internal {
        counter._value = 0;
    }
}

// File: IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: IRegistry.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.10;

interface IRegistry {
  event SystemWalletUpdated(address newWallet);
  event FeeVariablesChanged(uint256 indexed newFee, uint256 indexed newScale);
  event ContractStatusChanged(address indexed changed, bool indexed status);
  event CurrencyStatusChanged(address indexed changed, bool indexed status);

  function feeInfo(uint256 _salePrice) external view returns(address, uint256);
  function isPlatformContract(address toCheck) external view returns(bool);
  function isApprovedCurrency(address tokenContract) external view returns(bool);
  function setSystemWallet(address newWallet) external;
  function setFeeVariables(uint256 newFee, uint256 newScale) external;
  function setContractStatus(address toChange, bool status) external;
  function setCurrencyStatus(address tokenContract, bool status) external;
  function approveAllCurrencies() external;
}
// File: ISale.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.10;

interface ISale {
  struct Sale {
    uint256 id; // id of sale
    address owner; // address of NFT owner
    address nftContract;
    uint256 tokenId;
    uint256 amount; // amount of NFTs being sold
    uint256 purchased; // amount of NFTs purchased thus far
    uint256 startTime;
    uint256 endTime;
    uint256 price;
    uint256 maxBuyAmount;
    address currency; // use 0xaAaAaAaaAaAaAaaAaAAAAAAAAaaaAaAaAaaAaaAa for ETH
  }

  event NewSale(uint256 indexed id, Sale newSale);
  event SaleCancelled(uint256 indexed saleId);
  event Purchase(
    uint256 saleId, 
    address purchaser, 
    address recipient, 
    uint256 quantity
  );
  event NFTsReclaimed(uint256 indexed id, address indexed owner, uint256 indexed amount);
  event BalanceUpdated(
    address indexed accountOf, 
    address indexed tokenAddress, 
    uint256 indexed newBalance
  );

  function getSaleDetails(uint256 saleId) external view returns(Sale memory);
  function getSaleStatus(uint256 saleId) external view returns(string memory);
  function getClaimableBalance(address account, address token) external view returns(uint256);
  function createSale(
    address nftContract,
    uint256 tokenId,
    uint256 amount,
    uint256 startTime,
    uint256 endTime,
    uint256 price,
    uint256 maxBuyAmount,
    address currency
  ) external returns(uint256);
  function buy(uint256 saleId, address recipient, uint256 amountToBuy, uint256 amountFromBalance) external payable returns(bool);
  function claimNfts(uint256 saleId) external;
  function claimFunds(address tokenContract) external;
  function cancelSale(uint256 saleId) external;
}
// File: Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (access/Ownable.sol)

pragma solidity ^0.8.0;

import "./Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";
import "./Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: Sale.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.10;

/**
 * @title Sales contract for Etched NFTs
 * @author Linum Labs
 */

import "./Ownable.sol";
import "./SafeERC20.sol";
import "./Counters.sol";
import "./ReentrancyGuard.sol";
import "./ISale.sol";
import "./IRegistry.sol";

interface IEtchedNFT {
  function royaltyInfo(uint256 tokenId, uint256 _salePrice) external view returns(address, uint256);
  function balanceOf(address account, uint256 id) external view returns(uint256);
  function safeTransferFrom(address from, address to, uint256 id, uint256 amount, bytes memory data) external;
  function supportsInterface(bytes4 interfaceID) external returns(bool);
}

contract EtchedNFTSale is ISale, Ownable, ReentrancyGuard {
  using SafeERC20 for IERC20;
  using Counters for Counters.Counter;

  // already declared in ISale

  // struct Sale {
  //   uint256 id; // id of sale
  //   address owner; // address of NFT owner
  //   address nftContract;
  //   uint256 tokenId;
  //   uint256 amount; // amount of NFTs being sold
  //   uint256 purchased; // amount of NFTs purchased
  //   uint256 startTime;
  //   uint256 endTime;
  //   uint256 price;
  //   uint256 maxBuyAmount;
  //   address currency; // use 0xaAaAaAaaAaAaAaaAaAAAAAAAAaaaAaAaAaaAaaAa for ETH
  // }

  // address alias for using ETH as a currency
  address constant ETH = address(0xaAaAaAaaAaAaAaaAaAAAAAAAAaaaAaAaAaaAaaAa);

  Counters.Counter private _saleId;
  IRegistry private Registry;

  mapping(uint256 => Sale) private sales;
  mapping(uint256 => bool) private cancelled;
  mapping(uint256 => mapping(address => uint256)) private purchased;
  // user address => tokenAddress => amount
  mapping(address => mapping(address => uint256)) private claimableFunds;

  // already declared in ISale
  // event NewSale(uint256 indexed id, Sale indexed newSale);
  // event Purchase(
  //   uint256 saleId, 
  //   address purchaser, 
  //   address recipient, 
  //   uint256 quantity
  // );
  // event NFTsReclaimed(uint256 indexed id, address indexed owner, uint256 indexed amount);
  // event BalanceUpdated(
  //   address indexed accountOf, 
  //   address indexed tokenAddress, 
  //   uint256 indexed newBalance
  // );

  constructor(address registry) {
    Registry = IRegistry(registry);
  }

  /// @notice returns the details of a particular Sale
  /// @param saleId the index of the Sale to return the details of
  /// @return a Sale struct with the Sale details
  function getSaleDetails(uint256 saleId) external view returns(Sale memory) {
    require(saleId <= _saleId.current() && saleId > 0, "sale does not exist");
    return sales[saleId];
  }

  /// @notice returns the status of a particular Sale
  /// @dev possible statuses: PENDING, ACTIVE, ENDED, CANCELLED
  /// @param saleId the index of the Sale to return the status of
  /// @return an all-caps string of the status
  function getSaleStatus(uint256 saleId) public view override returns(string memory) {
    require(saleId <= _saleId.current() && saleId > 0, "sale does not exist");
    if(cancelled[saleId] || !Registry.isPlatformContract(address(this))) return "CANCELLED";
    if(block.timestamp < sales[saleId].startTime) return "PENDING";
    if(
      block.timestamp >= sales[saleId].startTime &&
      block.timestamp < sales[saleId].endTime &&
      sales[saleId].purchased < sales[saleId].amount
    ) return "ACTIVE";
    if(
      block.timestamp >= sales[saleId].endTime || 
      sales[saleId].purchased == sales[saleId].amount
    ) return "ENDED";
    revert("error");
  }

  /// @notice returns the user's balance of a particular token in the contract
  /// @dev use 0xaAaAaAaaAaAaAaaAaAAAAAAAAaaaAaAaAaaAaaAa for ETH
  /// @param account the address to query the balance of
  /// @param token the address of the token to query the account's balance of
  /// @return the uint256 balance of the account in the token queried
  function getClaimableBalance(address account, address token) external view returns(uint256){
    return claimableFunds[account][token];
  }

  /// @notice Creates a sale of NFTs from an approved contract
  /// @dev this custodies the NFTs in the contract until they are sold or reclaimed by the seller
  /// @dev use 0xaAaAaAaaAaAaAaaAaAAAAAAAAaaaAaAaAaaAaaAa for currency if ETH is being used
  /// @param nftContract the address of the contract the NFTs are from
  /// @param tokenId the index of the NFTs being sold in the NFT contract
  /// @param amount the amount of NFTs to sell
  /// @param startTime the uint256 Solidity timestamp the sale should start at
  /// @param endTime the uint256 Solidity timestamp the sale should end at
  /// @param maxBuyAmount the maximum amount a signle address can buy from the sale
  /// @param currency the address of the token NFTs can be bought in
  /// @return the uint256 ID of the sale
  function createSale(
    address nftContract,
    uint256 tokenId,
    uint256 amount,
    uint256 startTime,
    uint256 endTime,
    uint256 price,
    uint256 maxBuyAmount,
    address currency
  ) external nonReentrant returns(uint256) {
    IEtchedNFT NftContract = IEtchedNFT(nftContract);
    require(Registry.isPlatformContract(nftContract) == true, "NFT not in approved contract");
    require(Registry.isPlatformContract(address(this)) == true, "This contract is deprecated");
    require(Registry.isApprovedCurrency(currency) == true, "currency not supported");
    require(NftContract.supportsInterface(0x2a55205a), "contract must support ERC2981");
    require(NftContract.balanceOf(msg.sender, tokenId) >= amount, "insufficient NFT balance");
    require(endTime > startTime, "error in start/end params");
    require(maxBuyAmount > 0, "maxBuyAmount must be non-zero");
    _saleId.increment();
    uint256 saleId = _saleId.current();

    sales[saleId] = Sale({
      id: saleId,
      owner: msg.sender,
      nftContract: nftContract,
      tokenId: tokenId,
      amount: amount,
      purchased: 0,
      startTime: startTime,
      endTime: endTime,
      price: price,
      maxBuyAmount: maxBuyAmount,
      currency: currency
    });

    NftContract.safeTransferFrom(
      msg.sender, 
      address(this), 
      tokenId,
      amount,
      ""
    );

    emit NewSale(
      saleId,
      sales[saleId]
    );

    return saleId;
  }

  /// @notice purchases NFTs from a sale
  /// @dev system fees and royalties are automatically split off
  /// @param saleId the index of the sale to purchase from
  /// @param recipient the address the purchased NFTs should be sent to
  /// @param amountToBuy the amount of NFTs to purchase
  /// @param amountFromBalance the amount of claimable balance on this contract to use
  ///    towards the purchase
  /// @return a bool indicating success
  function buy(
    uint256 saleId, 
    address recipient, 
    uint256 amountToBuy,
    uint256 amountFromBalance
  ) external payable nonReentrant returns(bool) {
    require(Registry.isPlatformContract(address(this)) == true, "This contract is deprecated");
    require(keccak256(bytes(getSaleStatus(saleId))) == keccak256(bytes("ACTIVE")), "sale is not active");
    Sale memory currentSale = sales[saleId];
    require(purchased[saleId][msg.sender] + amountToBuy <= currentSale.maxBuyAmount, "buy quantity too high");
    require(amountToBuy <= currentSale.amount - currentSale.purchased,
      "not enough stock for purchase");
    address currency = currentSale.currency;
    require(amountFromBalance <= claimableFunds[msg.sender][currency],
      "not enough balance"
    );

    uint256 tokenId = currentSale.tokenId;

    IEtchedNFT Nft = IEtchedNFT(currentSale.nftContract);
    (address artistAddress, uint256 royalties) = Nft.royaltyInfo(tokenId, amountToBuy * currentSale.price);

    if(currency != ETH) {
      IERC20 Token = IERC20(currency);

      Token.safeTransferFrom(msg.sender, address(this), (amountToBuy * currentSale.price) - amountFromBalance);
    } else {
      require(msg.value == (amountToBuy * currentSale.price) - amountFromBalance, "msg.value + balance != price");
    }
    if(amountFromBalance > 0){
      claimableFunds[msg.sender][currency] -= amountFromBalance;
      emit BalanceUpdated(msg.sender, currency, claimableFunds[msg.sender][currency]);
    }

    // system fee
    (address systemWallet, uint256 fee) = Registry.feeInfo(amountToBuy * currentSale.price);
    claimableFunds[systemWallet][currency] += fee;
    emit BalanceUpdated(systemWallet, currency, claimableFunds[systemWallet][currency]);

    // artist royalty if artist isn't the seller
    if(currentSale.owner != artistAddress){
      claimableFunds[artistAddress][currency] += royalties;
      emit BalanceUpdated(artistAddress, currency, claimableFunds[artistAddress][currency]);
    } else {
      // since the artist is the seller
      royalties = 0;
    }

    // seller gains
    claimableFunds[currentSale.owner][currency] += (amountToBuy * currentSale.price) - fee - royalties;
    emit BalanceUpdated(currentSale.owner, currency, claimableFunds[currentSale.owner][currency]);

    sales[saleId].purchased += amountToBuy;
    purchased[saleId][msg.sender] += amountToBuy;

    Nft.safeTransferFrom(
      address(this), 
      recipient, 
      tokenId,
      amountToBuy, 
      ""
    );

    emit Purchase(saleId, msg.sender, recipient, amountToBuy);

    return true;
  }

  /// @notice claims unsold NFTs
  /// @dev this is for the original seller to reclaim unsold NFTs
  ///    from an ended or cancelled sale
  /// @param saleId the index of the sale to reclaim the NFTs from
  function claimNfts(uint256 saleId) external {
    bytes32 status = keccak256(bytes(getSaleStatus(saleId)));
    require(
      status == keccak256(bytes("CANCELLED")) || 
      status == keccak256(bytes("ENDED")),
      "cannot claim before sale closes"  
    );
    require(msg.sender == sales[saleId].owner,
      "only nft owner can claim"
    );
    require(sales[saleId].purchased < sales[saleId].amount, "stock already sold or claimed");

    uint256 stock = sales[saleId].amount - sales[saleId].purchased;
    sales[saleId].purchased = sales[saleId].amount;
    IEtchedNFT Nft = IEtchedNFT(sales[saleId].nftContract);
    Nft.safeTransferFrom(
      address(this),
      sales[saleId].owner, 
      sales[saleId].tokenId,
      stock,
      ""
    );

    emit NFTsReclaimed(saleId, msg.sender, stock);
  }

  /// @notice Explain to an end user what this does
  /// @dev use 0xaAaAaAaaAaAaAaaAaAAAAAAAAaaaAaAaAaaAaaAa for ETH
  /// @param tokenContract the address of the token being claimed
  function claimFunds(address tokenContract) external {
    require(claimableFunds[msg.sender][tokenContract] > 0, "nothing to claim");
    uint256 payout = claimableFunds[msg.sender][tokenContract];
    if(tokenContract != ETH) {
      IERC20 Token = IERC20(tokenContract);
      claimableFunds[msg.sender][tokenContract] = 0;
      Token.safeTransfer(msg.sender, payout);
    } else {
      claimableFunds[msg.sender][tokenContract] = 0;
      (bool success,) = msg.sender.call{ value: payout }("");
      require(success, "ETH payout failed");
    }
    emit BalanceUpdated(msg.sender, tokenContract, claimableFunds[msg.sender][tokenContract]);
  }

  /// @notice cancels a sale
  /// @dev can only be called by NFT owner or platform owner
  /// @param saleId the index of the sale to cancel
  function cancelSale(uint256 saleId) external {
    require(msg.sender == sales[saleId].owner || 
      msg.sender == owner(),
      "only owner or sale creator"  
    );
    cancelled[saleId] = true;

    emit SaleCancelled(saleId);
  }

  /// @notice callback functioning for signalling ERC1155 support
  /// @dev in order to receive a safeTransfer, contracts must return
  ///    `bytes4(keccak256("onERC1155Received(address,address,uint256,uint256,bytes)"))`
  ///    (0xf23a6e61) when queried
  /// @param operator The address which initiated the transfer (i.e. msg.sender)
  /// @param from The address which previously owned the token
  /// @param id The ID of the token being transferred
  /// @param value The amount of tokens being transferred
  /// @param data Additional data with no specified format
  /// @return `bytes4(keccak256("onERC1155Received(address,address,uint256,uint256,bytes)"))` if transfer is allowed
  function onERC1155Received(
    address operator, 
    address from,
    uint256 id, 
    uint256 value, 
    bytes memory data
    ) external pure returns(bytes4){
    // 0xf23a6e61 = bytes4(keccak256("onERC1155Received(address,address,uint256,uint256,bytes)")
    return 0xf23a6e61;
  }
}